Module 8

Customizing Printed Reports

In this module you will learn how to:

- Add headings and footings
- Change column titles
- Change field formats and justification
- Add totals

Supporting Files/Documents:

- FOCUS Documentation
- Active Current Status File

Headings and Footings

The HEADING command is used after the TABLE FILE command and allows you to supply text that will appear at the top of every page of a report. The syntax is:

HEADING or HEADING CENTER"text"

The FOOTING command is used before the END command and allows you to supply text that will appear at the bottom of every page of a report. The syntax is:

FOOTING or FOOTING CENTER "text"

- Each line of text can be up to 128 characters in length including the two double quotes.
- A single report can contain up to 57 lines of text.
- The optional command CENTER will center the text per the report output, not on the page.
- The command FOOTING will automatically give you one blank line after the report data.
- Use a spot-marker to insert blank lines, (e.g., "</2") or to embed the value for a data element (e.g., "DATA AS OF: <AC:DAO").
- The system variable for the current date can be supplied in headings or footings (e.g., "GENERATED: &DATE").

Report Request

```
\ensuremath{^{-*}} ALPHA LIST OF ACTIVE/ONLEAVE EMPLOYEES EX AC
```

TABLE FILE AC

HEADING CENTER

"THE CALIFORNIA STATE UNIVERSITY"

"ACTIVE/ONLEAVE EMPLOYEES AS OF <AC:DAO"

11 1

PRINT AC:CLASS
AC:FTE
AC:SALARY

BY AC: WNAME

BY AC:SSA NOPRINT

IF RECORDLIMIT EQ 10

FOOTING

"</1"

"REPORT RUN ON - &DATE"

"CREATED BY JANE GARCIA - PERSONNEL </1"

"CONFIDENTIAL DATA - PLEASE HANDLE ACCORDINGLY"

Report Generated

THE CALIFORNIA STATE UNIVERSITY ACTIVE/ONLEAVE EMPLOYEES AS OF 01/27/2006

AC:WNAME	CLS	AC:FTE	AC:SALARY
ARDVARK, CINDY A	3306	1.000	\$10,041.00
BEAR, CLARK F	2358	.200	\$1,275.00
FROG, NANCY B	3312	1.000	\$6,346.00
KOALA, JOHN L	2358	.600	\$3,276.00
LLAMA, JOAN M	2360	1.000	\$5,248.00
MONKEY, RON J	1038	1.000	\$2,917.00
O'HARE, JUSTIN J	2354	.267	\$713.33
	2354	.057	\$197.01
PENGUIN, ROSS	6526	1.000	\$4,211.00
YAK, SAMUEL C	3312	1.000	\$6 , 776.00

REPORT RUN ON - 02/01/2006

CREATED BY JAN GARCIA - PERSONNEL

CONFIDENTIAL DATA - PLEASE HANDLE ACCORDINGLY

Creating New Column Titles

Default column titles can be changed to a title you supply using the AS command. The syntax is:

fieldname AS 'new, title'

fieldname Can be a sort field, display (verb) field, column total or row

total

title Is the new column title enclosed in single quotation marks

- To eliminate column titles use AS '' (two single quotes without a space between). For example, PRINT AC: WNAME AS ''
- Column titles can have up to five lines of text (four commas).
- The AS command can be used to increase or decrease the width of your column titles.

Report Request:

```
EX AC
TABLE FILE AC
SUM AC:FTE AS 'TOTAL,FTE'
BY AC:SEX AS 'S,E,X'
BY AC:WNAME AS 'EMPLOYEE'
IF RECORDLIMIT EQ 5
END
```

S		
E		TOTAL
Χ	EMPLOYE	FTE
-		
F	FROG, NANCIE L	1.000
	MONKEY, GINA S	.600
M	BEAR, CLARK	1.000
	LLAMA, ROLAND J	1.000
	PENGUIN, RICHARD R	1.000

Changing Column Formats and Title Justification

FOCUS enables you to change the length of the fields and to specify whether column titles are left justified, right justified, or centered. By default, column titles for alphanumeric fields are left justified and column titles for numeric and date fields are right justified. The syntax is:

fieldname / justification / format

fieldname Can be a sort field, display (verb) field, column total or row total.

justification Can be the command R for right justification. L for left

justification and C for centered.

format Is an optional format specification **only** for display and

row total fields. If used, the type (A, D, I, P,) must remain the same, but the length can be changed.

- For ACROSS fields, data values, not column titles, are justified as specified.
- Display fields and row totals can be increased or decreased by specifying a new format length.

Report Request:

EX AC
TABLE FILE AC
SUM AC:FTE/P6.2
BY AC:SEX
BY AC:WNAME/C
IF RECORDLIMIT EQ 5
END

SEX	AC:WNAME	AC:FTE
F	FROG, NANCIE L	1.00
	MONKEY, GINA S	.60
M	BEAR, CLARK	1.00
	LLAMA, ROLAND J	1.00
	PENGUIN, RICHARD R	1.00

Column Totals and Row Totals

FOCUS enables you to summarize numeric information and produce totals for columns or rows of numbers in your reports. The syntax is:

ON TABLE COLUMN-TOTAL [AS 'name'] ON TABLE ROW-TOTAL [AS 'name']

- When you are summing fields with different formats, the default format is D12.2, otherwise, the format of the total field is the same format of the field.
- COLUMN-TOTAL and ROW-TOTAL can be used together to form a matrix. If used together, COLUMN-TOTAL must be first.
- Only verb fields are included in the column total.

Report Request:

EX AC

TABLE FILE AC

COUNT AC: POSITION

BY AC:CBID

BY AC:CLASS

ACROSS AC:SEX

IF AC:CBID EQ M80

ON TABLE COLUMN-TOTAL ROW-TOTAL

END

		SEX		
		F	M	TOTAL
CBID	CLS			
М80	3300	5	6	11
	3306	16	13	29
	3312	20	15	35
	3318	3	3	6
TOTAL		44	37	81

Subtotals

The command SUBTOTAL generates a subtotal on all numeric verb fields whenever the sort field changes value. The syntax is:

BY fieldname SUBTOTAL MULTILINES

- The command SUB-TOTAL (with a hyphen) can be used to produce a subtotal for the BY field and any higher-level sort fields when their values change.
- The MULTILINES command suppresses a subtotal line if there is only one value at a sort break and will automatically skip a line between rows.
- The default (fieldname) subtotal title can be changed by using the AS 'text' command.
- A grand total is automatically generated at the end of the report.

Report Request:

EX AC
TABLE FILE AC
PRINT AC:FTE
BY AC:WNAME
BY AC:SSA SUBTOTAL MULTILINES AS ' TOTAL FOR:'
IF RECORDLIMIT EQ 5
END

AC:WNAME	AC:SSA	AC:FTE
FROG, CHRISTOPHER	001-00-1111	1.000
GIRAFFE, RICHARD B	003-00-3333	.917
TOTAL FOR: 003-00-3333		1.000
HIPPO, CHARITY B	004-00-4444	1.000
MOUSE, NORMA K	002-00-2222	.440
TOTAL		3.400

Sample Report Request

The report request below illustrates all the commands covered in this module. Note that these commands are only useful for printed reports and will be ignored if the data is held to a hold file for downloading.

Report Request

```
-* ALPHA LIST OF ACTIVE/ONLEAVE EMPLOYEES
EX AC
TABLE FILE AC
HEADING CENTER
"THE CALIFORNIA STATE UNIVERSITY"
"ACTIVE/ONLEAVE EMPLOYEES AS OF <AC:DAO"
PRINT AC: WNAME/C/A25 AS 'EMPLOYEE'
     AC:CLASS AS 'CLASS'
AC:FTE AS 'FTE'
     AC:SALARY/P8.2M AS 'SALARY'
BY AC: WNAME NOPRINT
ON TABLE COLUMN-TOTAL
FOOTING
"</1"
"REPORT RUN ON - &DATE"
"CREATED BY JANE GARCIA - PERSONNEL </1"
"CONFIDENTIAL DATA - PLEASE HANDLE ACCORDINGLY"
```

Report Generated

THE CALIFORNIA STATE UNIVERSITY ACTIVE/ONLEAVE EMPLOYEES AS OF 01/27/2006

EMPLOYEE	CLASS	FTE	SALARY
ARDVARK, CINDY A	3306	1.000	\$10,041.00
BEAR, CLARK F	2358	.200	\$1,275.00
FROG, NANCY B	3312	1.000	\$6,346.00
KOALA, JOHN L	2358	.600	\$3,276.00
LLAMA, JOAN M	2360	1.000	\$5,248.00
MONKEY, RON J	1038	1.000	\$2,917.00
O'HARE, JUSTIN J	2354	.267	\$713.33
O'HARE, JUSTIN J	2354	.057	\$197.01
PENGUIN, ROSS	6526	1.000	\$4,211.00
YAK, SAMUEL C	3312	1.000	\$6,776.00
m			
TOTAL		7.124	\$41,000.34

REPORT RUN ON - 02/01/2006 CREATED BY JAN GARCIA - PERSONNEL

CONFIDENTIAL DATA - PLEASE HANDLE ACCORDINGLY

Practice

Modify the programs in your library as follows:

- 1. Add headings that provide a description of the report data.
- 2. Add footings containing reference data (e.g., report request name, contact information, etc.).
- 3. Change the default column titles to 'reader friendly' titles.
- 4. Eliminate excess space between the data columns by changing the column length.
- 5. Add totals (where applicable).
- 6. Add a small record limit (100 records or less) to your reports if needed.

After you have modified your reports, execute them using either online or batch execution.

What You Have Learned

In this module, you have learned how to customize printed reports by:

- Adding headings and footings
- Changing column titles
- Changing field formats and justification
- Adding column totals, row totals and subtotals